
Introduction to Conga

You need Conga to…
• make two Dyalog APL instances speak with each other
• make Dyalog APL speak to networked devices
• let Dyalog provide general services

All of the above may be…
• through a network

and/or
• on a single local machine

Introduction to Conga

Do not use Conga (directly) to…
• run a website → MiServer
• provide web services → SAWS
• call an APL function over a network → JSONServer
• fetch a file or do other HTTP actions → HttpCommand

Introduction to Conga

What is Conga?
• Layer between APL and OS sockets
• Implementation of protocols
• Multi-thread manager
• Object Oriented APL tool (new in Conga 3.0!)

Introduction to Conga

Sockets
• UDP – User Datagram Protocol
• TCP – Transmission Control Protocol
• Both are on top of the IP layer (Internet Protocol)

• TCP/IP is a tunnel between endpoints

Introduction to Conga

Sockets

Introduction to Conga

Sockets: How a connection begins

SERVER CLIENT

listens on socket bound to 
address and port

connects to address and port
receives new connection and

creates new socket for it

keeps listening for more 
connections on old socket

Introduction to Conga

Sockets: Endpoints
• IPv4 address, e.g. 81.94.205.35 (32 bit)

or
IPv6 address, e.g. 2a02:2658:1012::35 (128 bit)

• Port, e.g. 80 (16 bit)

• Server listens on socket bound to address and port
• Client connects to address and port
• Client has ephemeral (unknown) port (≥50000)

Introduction to Conga

Domain Name Service et al.
iConga.Clt 'DYA' 'www.dyalog.com' 'http'

┌─┬───┐
│0│DYA│
└─┴───┘

iConga.GetProp 'DYA' 'peeraddr'
┌─┬──────────────────────────────────────┐
│0│┌────┬───────────────┬────────────┬──┐│
│ ││IPv4│81.94.205.35:80│81 94 205 35│80││
│ │└────┴───────────────┴────────────┴──┘│
└─┴──────────────────────────────────────┘

iConga.GetProp 'DYA' 'localaddr'
┌─┬────────────────────────────────────────────────┐
│0│┌────┬────────────────────┬──────────────┬─────┐│
│ ││IPv4│192.168.17.128:54442│192 168 17 128│54442││
│ │└────┴────────────────────┴──────────────┴─────┘│
└─┴────────────────────────────────────────────────┘

Name → IP address Service → port

Introduction to Conga

Loading Conga
Load APL code

)copy conga Conga

Load C library
iConga←Conga.Init ''

Conga 3.0+: get events instead of errors
iConga.SetProp '.' 'EventMode' 1

Introduction to Conga

SERVER Session
)copy conga Conga
]load [DYALOG]/Samples/Conga/RPCServices/SimpleReverser -nolink 

sr1←Conga.Srv 5000 #.SimpleReverser 
sr1.Start

)wsid SERVER
]caption editor {WSID}    ⍝ Don't do this in RIDE!

Introduction to Conga

CLIENT Session
)copy Conga

)wsid CLIENT

iConga←Conga.Init   '' ⍝ load lib and create instance
iConga.SetProp   '.'   'eventmode'   1 ⍝ treat errors as events

iConga.Clt 'MyCon'   ''   5000 ⍝ new client connection
iConga.Send 'MyCon.MyCmd'   'repaid desserts' ⍝ send command
iConga.Wait 'MyCon.MyCmd'   1000 ⍝ wait for response
iConga.Close 'MyCon' ⍝ close connection

Introduction to Conga

Basic Command Connection

Follow the cheat sheet

Introduction to Conga

Recap — what did we just do?

iConga.Clt 'MyCon' '' 5000

AddressConnection name Port

Introduction to Conga

Syntax of Wait
iConga.Wait 'MyCon.MyCmd' 1000

┌─┬───────────┬───────┬─────────────────────────────────┐
│0│MyCon.MyCmd│Receive│┌───────────────┬───────────────┐│
│ │           │       ││repaid desserts│stressed diaper││
│ │           │       │└───────────────┴───────────────┘│
└─┴───────────┴───────┴─────────────────────────────────┘

Error code: Zero means "success" — result has this 4-element format

Object: ConnectionName.CommandName

Event that has occurred

Data for event

Timeout: How long we'll wait (ms)

Object: ConnectionName.CommandName

Introduction to Conga

Error-result of Wait (with 3.0+ setting)
iConga.SetProp '.' 'EventMode' 1
iConga.Wait 'MyCon.MyTypo' 1000

┌────┬────────────────────┬┐
│1010│ERR_OBJECT_NOT_FOUND││
└────┴────────────────────┴┘

Error code: Non-zero means failure — result has this 3-element format

Error type

Error description (may be '')

Introduction to Conga

SimpleReverser
:Class SimpleReverser : #.Conga.Connection

∇ MakeN arg
:Access Public
:Implements Constructor :Base arg

∇

∇ onReceive(obj data)
:Access public override
_←Respond obj(data(⌽data))

∇
:EndClass 

Introduction to Conga

Modify SimpleReverser
:Class SimpleReverserAddr : #.Conga.Connection

∇ MakeN arg;iConga;server
:Access Public
:Implements Constructor :Base arg
iConga←srv.LIB
⎕←'Mine: ',iConga.GetProp Name'LocalAddr'
⎕←'Theirs: ',iConga.GetProp Name'PeerAddr'

:For server :In iConga.Names '.'
⎕←'Server:',server,' Connections: ',iConga.Names server

:EndFor
∇
∇ onReceive(obj data)

:Access public override
⎕←'Command: ',obj
_←Respond obj(data(⌽data))

∇

Introduction to Conga

SERVER Session

]set workdir ,/<mypath>/
]save SimpleReverserAddr /<mypath>/ -makedir

sr1.Stop

sr1a←Conga.Srv 5000 #.SimpleReverserAddr
sr1a.Start

Introduction to Conga

CLIENT Session
iConga.Close 'MyCon'   ⍝ Did you close it?

(err con)←iConga.Clt 'MyCon' '' 5000

con SendReceive 'repaid desserts'

iConga.Close con

Introduction to Conga

Log Server Activity

∇ result←connection SendReceive text
1. Send a request to the server
2. Wait for a result
3. Handle all kinds of results

Keep an eye on the server's session
Don't forget to save your work!

Introduction to Conga

Syntax of Send
iConga.Send 'MyCon.MyCmd' 'repaid desserts'

┌─┬───────────┐
│0│MyCon.MyCmd│
└─┴───────────┘

Error code: Zero means "success"

Object: ConnectionName.CommandName

Data: Any APL array

Object: ConnectionName.CommandName

Introduction to Conga

Modify SimpleReverserAddr
:Class SimpleReverserAddrPro : #.Conga.Connection

∇ onReceive(obj data);i;n
:Access public override
⎕←'Command ',obj
⍝ Inform the waiting client on progress
:For i :In ⍳¯1+n←10

⎕DL 4
_←Progress obj(⎕←(⍕100×i÷n),'% done')

:EndFor
_←Respond obj(data(⌽data))

∇

Introduction to Conga

SERVER Session

]save SimpleReverserAddrPro /<mypath>/

sr1a.Stop

sr1b←Conga.Srv 5000 #.SimpleReverserAddrPro
sr1b.Start

Introduction to Conga

CLIENT Session
iConga.Close 'MyCon' ⍝ Did you close it?

(err con)←iConga.Clt 'MyCon' '' 5000

con SendReceive 'repaid desserts'
]save SendReceive /<mypath>/

iConga.Names '.' ⍝ list open connections
iConga.Close con

Introduction to Conga

r←con SendReceive data
(err cmd)←2↑res←iConga.Send con data
⍝ throw error if err is non-zero
:Repeat

err←⊃res←iConga.Wait cmd 1500
⍝ throw error if err is non-zero
(err obj evt dat)←4↑res
:Select evt
⍝ handle every event type
:EndSelect

:Until evt≡'Receive'

Introduction to Conga

Progress
• You can send any APL array back
• When Receive is called, all remaining Progress reports 

are discarded

Introduction to Conga

Our SendReceive
r←con SendReceive data;cmd;err;obj;evt;dat;res
(err cmd)←2↑res←iConga.Send con data
('Send failed ',∊⍕res)⎕SIGNAL(err≠0)/999
:Repeat

err←⊃res←iConga.Wait cmd 1500
('Wait failed ',∊⍕res)⎕SIGNAL(err≠0)/999
(err obj evt dat)←4↑res
:Select evt
:Case 'Timeout' ⋄ ⎕←'very bored...'
:Case 'Error' ⋄ ⎕←'ooooops: ',∊⍕dat
:Case 'Receive' ⋄ r←dat
:Case 'Progress' ⋄ ⎕←'so far: ',dat
:Else            ⋄ r←⍬
:EndSelect

:Until evt≡'Receive'

Introduction to Conga

Mode
• 'Command' (default, if omitted)
• 'Text' and 'Raw'

• 'BlkText' and 'BlkRaw' (thin protocol)
• 'http' (→MiServer/JSONServer/HttpCommand)

• 'Udp' (broadcasts with no guarantee — WIP)

Introduction to Conga

General Events (can happen in all modes)
• Connect
• Error
• Timeout
• Closed

Introduction to Conga

Command Mode
• Implements an APL specific protocol
• APL at both ends
• Possible events: Receive, Progress
• Object trees:

Server—⪪Connection—⪪Command—⪪Progress
Client—⪪Command—⪪Progress

(You may have multiple children for every —⪪)

Introduction to Conga

'Text' Mode (applies to 'Raw' mode too)
• No protocol
• Can have anything on the other end
• MaxBlockSize
• Both sides use Send
• Possible events: Block, BlockLast
• Object trees:

Server—⪪Connection—⪪Block
Client—⪪Block

• Optional: "End Of Message" ('EOM')

Introduction to Conga

:Class SimpleText : #.Conga.Connection
∇ MakeN arg
:Access Public
:Implements Constructor :Base arg

∇
∇ sa←ServerArgs
:Access public shared
sa←('Mode' 'Text')('BufferSize' 1024)

∇
∇ onBlock(obj data)
:Access public
_←Send(⌽data)0

∇
∇ onBlockLast(obj data)
:Access public
Close Name

∇
:EndClass

Si
m

p
le

Te
xt

Introduction to Conga

SERVER Session

]save SimpleText /<mypath>/

sr1b.Stop

st2←Conga.Srv 5001 #.SimpleText
st2.Start

Introduction to Conga

CLIENT Session
(err con)←iConga.Clt  'MyTxt'  ''  5001  'text'  1024

(err blk)←iConga.Send  'MyTxt'  'repaid desserts'

(err obj evt dat)←4↑⎕←iConga.Wait con 1000

iConga.Close con

Introduction to Conga

Basic Text Connection

Follow the cheat sheet

Introduction to Conga

CLIENT
(err con)←iConga.Clt  'MyTxt'  ''  5001  'text' 1024

∇ sa←ServerArgs
:Access public shared
sa←('mode' 'text')('BufferSize' 1024)

∇

SERVER

Introduction to Conga

:Class SimpleTextEOM : #.Conga.Connection
∇ MakeN arg
:Access Public
:Implements Constructor :Base arg

∇
∇ sa←ServerArgs
:Access public shared
sa←('mode' 'text')('BufferSize' 1024)('EOM' '$STOP')

∇
∇ onBlock(obj data)
:Access public
_←Send(⌽data)0

∇
∇ onBlockLast(obj data)
:Access public
Close Name

∇
:EndClass

Si
m

p
le

Te
xt

EO
M

Introduction to Conga

SERVER Session

]save SimpleTextEOM mypath/

st2.Stop

st2a←Conga.Srv 5001 #.SimpleTextEOM
st2a.Start

Introduction to Conga

CLIENT Session
(err con)←iConga.Clt  'MyTxt'  ''  5001  'text'  1024

(err blk)←iConga.Send  'MyTxt'  'repaid desserts$STOPraw live$STOP'

(err obj evt dat)←4↑⎕←iConga.Wait con 1000
(err obj evt dat)←4↑⎕←iConga.Wait con 1000

iConga.Close con

Introduction to Conga

Segmented Messages

Follow the cheat sheet

Introduction to Conga

:Class SimpleTextSecure : #.Conga.Connection
∇ sa←ServerArgs;iConga;path;file;cert
:Access public shared
iConga←#.Conga.Init''

path←⎕SE.SALTUtils.DYALOG,'/TestCertificates/'
file←path,'server/localhost'
cert←⊃iConga.ReadCertFromFile file,'-cert.pem'
cert.KeyOrigin←'DER' (file,'-key.pem')

sa←('mode' 'text')('BufferSize' 1024)
sa,←('X509'cert)('SSLValidation'64)

_←iConga.SetProp'.' 'RootCertDir'(path,'ca/')
∇

Si
m

p
le

Te
xt

Se
cu

re

Introduction to Conga

SERVER Session

]save SimpleTextSecure mypath/

st2a.Stop

st2b←Conga.Srv 5005 #.SimpleTextSecure
st2b.Start

Introduction to Conga

CLIENT Session
anon←⎕NEW iConga.X509Cert

iConga.SetProp '.'  'RootCertDir' (⎕SE.SALTUtils.DYALOG,'/TestCertificates/ca/')

iConga.Clt 'MyTxt'  'localhost' 5005 'text' 1024 ('X509' anon) ('SSLValidation' 64)

iConga.Send 'MyTxt'  'repaid desserts'
iConga.Wait 'MyTxt' 1000

iConga.Close 'MyTxt'

Introduction to Conga

Secure Connection

Follow the cheat sheet

Introduction to Conga

Summary of Conga Functions
Init
Srv
Clt
Send
Wait
Respond
Progress

Introduction to Conga

How we used Conga functions
SERVER

Srv
Send
Respond
Progress

CLIENT
Init
Clt
Send

Wait

Introduction to Conga

Recap — Questions?
1 Basic Command Connection
1a Log Server Activity
1b Report Progress

2 Basic Text Connection
2a Segmented Messages
2b Secure Connection


